

Anexo D: Código de análisis de elementos finitos en software ANSYS Mechanical APDL (Termo mecánico acoplado) para cavidad base.

A continuación, se presenta, paso por paso, el modelamiento tridimensional (3D) del impacto usando APDL (ANSYS Parametric Design Language), este es un poderoso lenguaje de secuencias de comandos estructurado que se utiliza para interactuar con el solucionador ANSYS Mechanical. En el cuadro **D.1**, se presentan la configuración inicial del modelo, el cual obedece a una naturaleza acoplada multi física o termo estructural y la ubicación de los archivos de trabajo.

Cuadro D.1: Configuración inicial

```
! Thermo-Structural analysis
!
! https://www.padtinc.com/blog/starting-ansys-products-from-the-command-line/
! "C:\Program Files\ANSYS Inc\ANSYS Student\v212\ansys\bin\winx64\ansys2021R2.exe" -b -i ../TherStruc-
ANSYS.mac -o AnsysLog.out
!

!finish
!/clear

/TITLE, Thermo-Structural
/FILNAME,TherStruct,0           ! This sets the jobname to 'Structural'


/UDOC,1,DATE,0                 ! LOGO OFF

! REVERSE VIDEO (White background)
/RGB,INDEX,100,100,100, 0
/RGB,INDEX, 80, 80, 80,13
/RGB,INDEX, 60, 60, 60,14
/RGB,INDEX, 0, 0, 0,15
/REPLOT
```

Antes de construir un modelo, se debe seleccionar el tipo de elementos a usar y los modelos de material que mejor representan el sistema físico. Después de seleccionar esto, el modelo está listo para ser construido, como se detalla a continuación.

Se configuran las propiedades del material, cómo lo son módulo de Young, conductividad térmica, coeficiente alfa (dilatación), módulo de Poisson, la condición de isotropía y linealidad entre otros. El elemento SOLID5 tiene una capacidad de campo magnético, térmico, eléctrico, piezoeléctrico y estructural tridimensional con acoplamiento limitado entre los campos, tiene 8 nodos cada uno de ellos con tres grados de libertad en desplazamientos (x,y,z) y otros tres para la temperatura, voltaje y magnetismo, es un elemento finito hexaédrico, en la Fig. **D1** se presenta la geometría, sistema coordenado y grados de libertad de este elemento.

Fig. D.1. Geometría del elemento finito SOLID 5 en 3D.



El modelo solo debe contener elementos SOLID5 ya que no permite mezclar diferentes tipos de elementos explícitos 3-D en el mismo modelo. Este elemento se configura así (ver Cuadro **D.2**).

Cuadro D.2: Condiciones de entrada y preprocesamiento. Propiedades material y tipo de elemento.

```
!*-----
!* Inputs
!*-----

!* Material properties
ct = 130                ! thermal conductivity of aluminum (W/m.K)
alpha = 23.3e-6         ! CET [1/°K]
Tamb = 300              ! Temperatura ambiente [°K]
E = 7.1e10              ! Modulo de Young
nu = 0.33               ! Modulo de Poisson

!*-----
!* Preprocessor
!*-----
/PREP7                  ! Enter preprocessor

!* Element definition: https://www.mm.bme.hu/~gyebro/files/ans\_help\_v182/ans\_elem/Hlp\_E\_SOLID5.html
ET,1,SOLID5             ! 3-D 8-Node Structural Solid. 3 DOFs per node
KEYOPT,1,1,0            ! Element degrees of freedom: UX, UY, UZ, TEMP, VOLT, MAG
KEYOPT,1,3,1            ! Extra shapes: Do not include extra shapes
KEYOPT,1,5,0            ! Extra element output: Basic element printout

!* Material model (Isotropic)
MP,KXX,1,ct             ! Thermal conductivity
MP,EX,1,E               ! Elastic moduli (also EY, EZ)
MP,PRXY,1,NU            ! Major Poisson's ratios (also PRYZ, PRXZ)
MP,REFT,1,TAMB          ! Reference temperature. Must be defined as a constant;
MP,ALPX,1,ALPHA         ! Secant coefficients of thermal expansion (also ALPY, ALPZ).
```

Se genera la configuración del mallado del dominio del problema, el material relacionado y sus condiciones, la lectura de los datos de la malla, la cual está compuesta por el número de nodos, la ubicación (coordenadas) de los mismos en el espacio 3D y su conectividad. Estos datos de malla generalmente se encuentran en archivos de texto plano, para el caso de trabajo específico de esta tesis se tienen archivos CSV, archivo de valores separados por comas de Microsoft Excel. Posteriormente se ubican los comandos para graficar los nodos NPLLOT y elementos EPLLOT con base a la información cargada, en el cuadro **D.3** se presentan las líneas de código de lo anteriormente explicado.

Cuadro D.3: Selección y configuración de mallado e importación datos de malla

!* Mesh setup		
TYPE,1	! TYPE, ITYPE.	Sets the element type attribute pointer.
MAT,1	! MAT, MAT.	Sets the element material attribute pointer.
ESYS, pointer.	! ESYS, KCN.	Sets the element coordinate system attribute
SECNUM,	! SECNUM,SECID.	Sets the element section attribute pointer.
SHPP,OFF		
! READ MESH DATA		
NREAD,nodes_a.csv	! Read NODAL data.	
EREAD,elements_a.csv	! Read ELEMENT data	
! Plot mesh		
NPLOT		
EPLOT		
! WRITE MESH DATA		
!NWRITE		
!EWRITE		

Con los datos de entrada y malla configurada, se proceden a aplicar las cargas físicas y/o multi físicas del problema en las ubicaciones específicas del CAD del problema, para ello es vital conocer los nodos y su ubicación (coordenadas) para poder aplicar las diferentes cargas térmicas y mecánicas a las cuales está expuesto constantemente el proceso de moldeo por inyección. En el cuadro **D.4** se presentan las condiciones de frontera, los valores y su ubicación en las coordenadas Z, y en el cuadro **D.5** para las coordenadas X, la restricción de los grados de libertad en la estructura lo cual asimila a un empotramiento del molde en el sistema de inyección.

Cuadro D.4: Ubicación condiciones de frontera coordenadas Z.

! BOUNDARY CONDITIONS (BCs)
!NSEL,Type,Item,Comp,VMIN,VMAX,VINC,KABS
! D, Node, Lab, VALUE, VALUE2, NEND, NINC, Lab2, Lab3, Lab4, Lab5, Lab6, MESHFLAG: Defines degree-of-freedom constraints at nodes.

! -Z face

NSEL,S,LOC,Z,-0.016

! Select nodes at Z=-0.016

NPLOT

D,ALL,TEMP,300

! Restringir todos los nodos seleccionado y todos los GDLs

D,ALL,UX,0

! Restringir todos los nodos seleccionados (UX=0)

D,ALL,UY,0

! Restringir todos los nodos seleccionados (UY=0)

D,ALL,UZ,0

! Restringir todos los nodos seleccionados (UZ=0)

NSEL,ALL

! Reselect all nodes

NPLOT

! +Z face

NSEL,S,LOC,Z,0.05456

! Select nodes at Z=0.05456

NPLOT

D,ALL,TEMP,300

! Restringir todos los nodos seleccionado y todos los GDLs

D,ALL,UX,0

! Restringir todos los nodos seleccionados (UX=0)

D,ALL,UY,0

! Restringir todos los nodos seleccionados (UY=0)

D,ALL,UZ,0

! Restringir todos los nodos seleccionados (UZ=0)

NSEL,ALL

! Reselect all nodes

NPLOT

Cuadro D.5: Ubicación condiciones de frontera coordenadas X.

! -X

NSEL,S,LOC,X,-0.0396

! Select nodes at X=-0.0396

NPLOT

D,ALL,TEMP,300

! Restringir todos los nodos seleccionado y todos los GDLs

D,ALL,UX,0

! Restringir todos los nodos seleccionados (UX=0)

D,ALL,UY,0

! Restringir todos los nodos seleccionados (UY=0)

D,ALL,UZ,0

! Restringir todos los nodos seleccionados (UZ=0)

NSEL,ALL

! Reselect all nodes

NPLOT

! +X

NSEL,S,LOC,X,0.0396

! Select nodes at X=0.0396

NPLOT

D,ALL,TEMP,300

! Restringir todos los nodos seleccionado y todos los GDLs

D,ALL,UX,0

! Restringir todos los nodos seleccionados (UX=0)

D,ALL,UY,0

! Restringir todos los nodos seleccionados (UY=0)

D,ALL,UZ,0

! Restringir todos los nodos seleccionados (UZ=0)

NSEL,ALL

! Reselect all nodes

NPLOT

La temperatura del material del herramental y de fundido del material en el proceso de

inyección y la carga mecánica representada por la fuerza de cierre de la máquina (para mayor explicación diríjase al apartado 3.2.1).

Cuadro. D.6: Ubicación condiciones de frontera térmicas y mecánicas en coordenadas Y.

```
! Bottom face
NSEL,S,LOC,Y,-0.0392      ! Select nodes at Y=-0.0392
NPLOT
D,ALL,TEMP,300            ! Restringir todos los nodos seleccionados al valor indicado
D,ALL,UX,0                ! Restringir todos los nodos seleccionados (UX=0)
D,ALL,UY,0                ! Restringir todos los nodos seleccionados (UY=0)
D,ALL,UZ,0                ! Restringir todos los nodos seleccionados (UZ=0)
NSEL,ALL                  ! Reselect all nodes
NPLOT

! Mold face
NSEL,S,LOC,Y,-5.89E-04    ! Select nodes at Y=-0.0006
NPLOT
D,ALL,TEMP,503
F,ALL,FY,-37148.93        ! LOADS [N]
NSEL,ALL
NPLOT

! Writes load and load step option data to a file. The load step file will be named Jobname.Sn
LSWRITE, ,

FINISH                    ! Exit from /solu
```

Con todas las condiciones de preproceso, las condiciones de frontera y definición del problema multi físico acoplado, se muestra a continuación en el cuadro **D.7** los comandos para el cálculo y solución del problema, el post proceso, de tal forma que presente las variables de respuesta temperatura y desplazamientos en cada nodo de la estructura. Finalmente, el cuadro **D.8** muestra los comandos para generación de los archivos de texto que contienen la información de los resultados para cada variable respuesta (temperaturas y desplazamientos).

Cuadro D.7: Comandos para cálculo y solución del problema.

```
!*-----
!* Solver
!*-----
/SOLU                                ! Enters the solution processor.

!* Solver setup
ANTYPE,STATIC                        ! Perform a static analysis.
SOLVE                                ! Solution
FINISH                               ! Exit from /solu

!*-----
!* POST-PROCESSING
!*-----
/POST1

!* TEMPERATURES
/EFACET,1
PLNSOL, TEMP,, 0
!

!* DISPLACEMENTS
/EFACET,1
PLNSOL, U,SUM, 0,1.0
```

Cuadro. D.8: Exportación de variables.

```
!*-----
!* EXPORT VARIABLES
!*-----

!* TEMPERATURES
*VGET,NODAL_TEMP,NODE, ,TEMP,, ,2
*CREATE,ansuitmp
*CFOPEN,'TEMP_ansys','dat',' '
*VWRITE,NODAL_TEMP(1),,, , , , , ,
(G)
*CFCLOS
*END
/INPUT,ansuitmp
!*

!* DISPLACEMENTS
*VGET,nodal_solX,NODE, ,U,X, , ,2
*VGET,nodal_solY,NODE, ,U,Y, , ,2
*VGET,nodal_solZ,NODE, ,U,Z, , ,2
*CREATE,ansuitmp
*CFOPEN,'USUM_ansys','dat',' '
*VWRITE,nodal_solX(1),nodal_solY(1),nodal_solZ(1), , , , , ,
(G)
*CFCLOS
*END
/INPUT,ansuitmp
!*
```